


Lean4 を用いた Gödel の 第一/第二不完全性定理の形式化

齋藤彰悟

東北大学大学院理学研究科数学専攻

2024-12-05

0.1 FFL

形式論理学の形式化プロジェクト : github.com/FormalizedFormalLogic 

- 古典命題論理 `L0.Propositional`
- 直観主義命題論理 `L0.IntProp`
- 古典命題様相論理 `L0.Modal`
- **古典一階述語論理 `L0.FirstOrder`**
- 直観主義一階述語論理 `L0.IntFO`

0.2 不完全性定理

理論 T を計算可能な一階算術の理論とする.

定理 0.2.1 (不完全性定理):

- (G1) T が基礎的な算術を扱える程度に強く, まともならば, T から証明も反証もできない論理式が存在する.
- (G2) T が十分強く, 無矛盾ならば, T の無矛盾性を表す文は証明できない.

1986(!), Shanker Nqthm による G1 の形式化.

2004, O'Connor Coq による G1 の形式化.

2004, Harrison HOL Light による G1 の形式化.

0.2 不完全性定理

2021, Paulson Isabelle/HOL による $G1$ と $G2$ の形式化 [1]. 算術ではなく遺伝的有限集合の理論 HF (ペアノ算術 PA と同等) の上で証明している.

- 私が知る限り唯一の第二不完全性定理の形式化¹.

定理 0.2.2 (Formalized by Paulson):

(G1) HF からは証明も反証もできない論理式が存在する.

(G2) 自己の無矛盾性を表す文は HF からは証明できない.

¹ただし可証性条件を仮定した上での $G2$ の証明はいくつか存在する.

0.2 不完全性定理

次の不完全性定理の一つのバリエーションを形式化した¹：

定理 0.2.3:

- (G1) T が R_0 より強く Σ_1 -健全ならば, T から証明も反証もできない論理式が存在する.
- (G2) T が $I\Sigma_1$ より強く無矛盾ならば, T の無矛盾性を表す文は証明できない.

- Σ_1 -健全 : T から証明可能な Σ_1 文は標準モデルの上で真.
- R_0 : Cobham の最弱の算術.
- $I\Sigma_1$: ペアノ算術の断片理論.

¹後述するように完全性定理を用いているため非構成的.

0.2 不完全性定理

以下の5つのステップを踏んで証明を行う.

1. 一階述語論理の形式化.
 - 項と論理式, 代入操作などの形式化.
 - 証明可能性, 充足性の形式化.
2. 完全性定理.
 - 証明探索
3. $I\Sigma_1$ の内部で算術を展開する.
 - 指数関数が定義可能であることを証明する.
 - 有限集合, 有限列といった基礎概念をコード化する.
4. メタ数学の算術化.
 - 項, 論理式, 証明可能性などをコード化する.
5. Hilbert-Bernays-Löb の可証性条件

1. 一階述語論理の形式化.

1.1 項・論理式の形式化

論理式（擬論理式）は常に否定標準形を取るよう¹に定義する。すなわち,

$$\varphi, \psi ::= \top \mid \perp \mid R(\vec{v}) \mid \neg R(\vec{v}) \mid \varphi \wedge \psi \mid \varphi \vee \psi \mid \forall \varphi \mid \exists \varphi$$

```
inductive Semiformula (L : Language) (ξ : Type*) : ℕ → Type _ where
| verum   {n} : Semiformula L ξ n
| falsum  {n} : Semiformula L ξ n
| rel     {n} : {arity : ℕ} → L.Rel arity → (Fin arity → Semiterm L ξ n) → Semiformula L ξ n
| nrel    {n} : {arity : ℕ} → L.Rel arity → (Fin arity → Semiterm L ξ n) → Semiformula L ξ n
| and     {n} : Semiformula L ξ n → Semiformula L ξ n → Semiformula L ξ n
| or      {n} : Semiformula L ξ n → Semiformula L ξ n → Semiformula L ξ n
| all     {n} : Semiformula L ξ (n + 1) → Semiformula L ξ n
| ex      {n} : Semiformula L ξ (n + 1) → Semiformula L ξ n
```

論理式

$$\overline{98 + 6748 = 6846}$$

$$(\forall x)(\forall y)[a \cdot (x + y) = a \cdot x + a \cdot y] \quad \text{“} a \mid \forall x y, a * (x + y) = a * x + a * y \text{”}$$

$$(\forall x)[x < \bar{n} \leftrightarrow \bigvee_{i < n} x = \bar{i}]$$

糖衣構文

$$\text{“} 98 + 6748 = 6846 \text{”}$$

$$\text{“} \forall x, x < \uparrow n \leftrightarrow \forall i < n, x = \uparrow i \text{”}$$

¹否定 \neg は論理式の関数として定義する。また $\varphi \rightarrow \psi := \neg \varphi \vee \psi$ とする。

1.2 証明可能性の形式化

証明体系には Tait 計算を採用する.

$$\begin{array}{c}
 \frac{}{R(\vec{v}), \neg R(\vec{v}), \Delta} \text{axL} \quad \frac{}{\top, \Delta} \text{verum} \quad \frac{\varphi, \psi, \Delta}{\varphi \vee \psi, \Delta} \text{or} \quad \frac{\varphi, \Delta \quad \psi, \Delta}{\varphi \wedge \psi, \Delta} \text{and} \quad \frac{\varphi^+(\&0), \Delta^+}{\forall \varphi, \Delta} \text{all}^1 \quad \frac{\varphi(t), \Delta}{\exists \varphi, \Delta} \text{ex} \quad \frac{\Delta}{\Gamma} \text{wk} \quad \frac{\varphi, \Delta \quad \neg \varphi, \Delta}{\Delta} \text{cut} \quad \frac{}{\varphi} \text{root} \\
 \text{(if } \Delta \subseteq \Gamma) \quad \text{(if } \varphi \in T)
 \end{array}$$

```

inductive Derivation (T : Theory L) : Sequent L → Type _
| axL (Δ) {k} (r : L.Rel k) (v) : Derivation T (rel r v :: nrel r v :: Δ)
| verum (Δ) : Derivation T (τ :: Δ)
| or {Δ p q} : Derivation T (p :: q :: Δ) → Derivation T (p ∨ q :: Δ)
| and {Δ p q} : Derivation T (p :: Δ) → Derivation T (q :: Δ) → Derivation T (p ∧ q :: Δ)
| all {Δ p} : Derivation T (Rew.free.hom p :: Δ+) → Derivation T ((∀' p) :: Δ)
| ex {Δ p} (t) : Derivation T (p/[t] :: Δ) → Derivation T ((∃' p) :: Δ)
| wk {Δ Γ} : Derivation T Δ → Δ ⊆ Γ → Derivation T Γ
| cut {Δ p} : Derivation T (p :: Δ) → Derivation T (¬p :: Δ) → Derivation T Δ
| root {p} : p ∈ T → Derivation T [p]

```

Tait 計算は推論規則が少ないため扱いやすく, LK などの他の推論規則に比べ様々な証明が簡易になる (こともある).

¹ここでは自由変数を &0, &1, ... と表記する. また φ^+, Γ^+ はそれぞれに含まれる自由変数をインクリメントしたもの

2. 完全性定理

2.1 完全性定理

定理 2.1.1 (完全性定理): すべての 論理式 φ について,

$$T \vdash \varphi \iff T \models \varphi$$


\implies 方向 (健全性定理) は証明に関する帰納法により従う. \impliedby を示すには以下を証明すれば良い.

補題 2.1.1: 推件 Γ について, 次のいずれかが成立する.

- $T \vdash_{\Gamma} \Gamma$
- すべての $\varphi \in \Gamma$ を充足しないような T のモデルが存在する.

2.2 証明探索

証明は Tait 計算の証明探索による (参考: [2]).



```

inductive Reduc (T : Theory L) : Code L → Sequent L → Sequent L → Prop
| axlRefl {Γ : Sequent L} {k} (r : L.Rel k) (v) :
| Semiformula.rel r v ∉ Γ ∨ Semiformula.nrel r v ∉ Γ → Reduc T (Code.axL r v) Γ Γ
| verumRefl {Γ : Sequent L} : T ∉ Γ → Reduc T (Code.verum) Γ Γ
| and₁ {Γ : Sequent L} {p q : SyntacticFormula L} : p ∧ q ∈ Γ → Reduc T (Code.and p q) (p :: Γ) Γ
| and₂ {Γ : Sequent L} {p q : SyntacticFormula L} : p ∧ q ∈ Γ → Reduc T (Code.and p q) (q :: Γ) Γ
| andRefl {Γ : Sequent L} {p q : SyntacticFormula L} : p ∧ q ∉ Γ → Reduc T (Code.and p q) Γ Γ
| or {Γ : Sequent L} {p q : SyntacticFormula L} : p ∨ q ∈ Γ → Reduc T (Code.or p q) (p :: q :: Γ) Γ
| orRefl {Γ : Sequent L} {p q : SyntacticFormula L} : p ∨ q ∉ Γ → Reduc T (Code.or p q) Γ Γ
| all {Γ : Sequent L} {p : SyntacticSemiformula L 1} : ∀' p ∈ Γ → Reduc T (Code.all p) (p/[&(newVar Γ)] :: Γ) Γ
| allRefl {Γ : Sequent L} {p : SyntacticSemiformula L 1} : ∀' p ∉ Γ → Reduc T (Code.all p) Γ Γ
| ex {Γ : Sequent L} {p : SyntacticSemiformula L 1} {t : SyntacticTerm L} :
|   ∃' p ∈ Γ → Reduc T (Code.ex p t) (p/[t] :: Γ) Γ
| exRefl {Γ : Sequent L} {p : SyntacticSemiformula L 1} {t : SyntacticTerm L} :
|   ∃' p ∉ Γ → Reduc T (Code.ex p t) Γ Γ
| id {Γ : Sequent L} {p : SyntacticFormula L} (hp : p ∈ T) : Reduc T (Code.id p) ((¬∀p) :: Γ) Γ
| idRefl {Γ : Sequent L} {p : SyntacticFormula L} (hp : p ∉ T) : Reduc T (Code.id p) Γ Γ

local notation:25 Δ₁ " <[" c:25 "]" " Δ₂:80 ⇒ Reduc T c Δ₁ Δ₂
inductive ReducNat (T : Theory L) (s : ℕ) : Sequent L → Sequent L → Prop
| redex {c : Code L} : decode s.unpair.1 = some c → ∀ {Δ₂ Δ₁}, Reduc T c Δ₂ Δ₁ → ReducNat T s Δ₂ Δ₁
| refl : decode (α := Code L) s.unpair.1 = none → ∀ Δ, ReducNat T s Δ Δ

local notation:25 Δ₁ " <[" s:25 "]" " Δ₂:80 ⇒ ReducNat T s Δ₁ Δ₂

```

探索木が well-founded ならば証明可能.

```

noncomputable def syntacticMainLemma
  (wf : WellFounded (SearchTree.Lt T Γ))
  (p : SearchTree T Γ) :
  T ⇒ p.seq

```

探索木が well-founded でないならば Γ の反例モデル Model $T \Gamma$ が構成できる.

```

lemma Model.models : Model T Γ ⊨m* T

```

```

lemma semanticMainLemmaTop
  (nwf : ¬WellFounded (SearchTree.Lt T Γ))
  {φ : SyntacticFormula L}
  (h : φ ∈ Γ) :
  ¬Evalf (Model.structure T Γ) Semiterm.fvar φ

```

3. $\text{I}\Sigma_1$ の内部で算術を展開する

3.1 算術の公理系を定義する

PA⁻

```

inductive PeanoMinus : Theory  $\mathcal{L}_{or}$ 
| equal      :  $\forall \varphi \in \mathbf{EQ}, \text{PeanoMinus } \varphi$ 
| addZero    :  $\text{PeanoMinus } "x \mid x + 0 = x"$ 
| addAssoc   :  $\text{PeanoMinus } "x \ y \ z \mid (x + y) + z = x + (y + z)"$ 
| addComm    :  $\text{PeanoMinus } "x \ y \mid x + y = y + x"$ 
| addEqOfLt  :  $\text{PeanoMinus } "x \ y \mid x < y \rightarrow \exists z, x + z = y"$ 
| zeroLe     :  $\text{PeanoMinus } "x \mid 0 \leq x"$ 
| zeroLtOne  :  $\text{PeanoMinus } "0 < 1"$ 
| oneLeOfZeroLt :  $\text{PeanoMinus } "x \mid 0 < x \rightarrow 1 \leq x"$ 
| addLtAdd   :  $\text{PeanoMinus } "x \ y \ z \mid x < y \rightarrow x + z < y + z"$ 
| mulZero    :  $\text{PeanoMinus } "x \mid x * 0 = 0"$ 
| mulOne     :  $\text{PeanoMinus } "x \mid x * 1 = x"$ 
| mulAssoc   :  $\text{PeanoMinus } "x \ y \ z \mid (x * y) * z = x * (y * z)"$ 
| mulComm    :  $\text{PeanoMinus } "x \ y \mid x * y = y * x"$ 
| mulLtMul   :  $\text{PeanoMinus } "x \ y \ z \mid x < y \wedge 0 < z \rightarrow x * z < y * z"$ 
| distr      :  $\text{PeanoMinus } "x \ y \ z \mid x * (y + z) = x * y + x * z"$ 
| ltIrrefl   :  $\text{PeanoMinus } "x \mid x \not< x"$ 
| ltTrans    :  $\text{PeanoMinus } "x \ y \ z \mid x < y \wedge y < z \rightarrow x < z"$ 
| ltTri      :  $\text{PeanoMinus } "x \ y \mid x < y \vee x = y \vee x > y"$ 

```

3.1 算術の公理系を定義する

$I\varphi$	<pre>def succInd {ξ} (φ : Semiformula L ξ 1) : Formula L ξ := “!φ 0 → (∀ x, !φ x → !φ (x + 1)) → ∀ x, !φ x”</pre>
	<pre>def indScheme (Γ : Semiformula L ℕ 1 → Prop) : Theory L := { q ∃ p : Semiformula L ℕ 1, Γ p ∧ q = succInd p }</pre>
$I\Sigma_k$	<pre>abbrev indH (Γ : Polarity) (k : ℕ) : Theory \mathcal{L}_{or} := PA⁻ + indScheme \mathcal{L}_{or} (Arith.Hierarchy Γ k) abbrev iSigma (k : ℕ) : Theory \mathcal{L}_{or} := IND Σ k</pre>
PA	<pre>abbrev peano : Theory \mathcal{L}_{or} := PA⁻ + indScheme \mathcal{L}_{or} Set.univ</pre>
R_0	<pre>inductive CobhamR0 : Theory \mathcal{L}_{or} equal : ∀ φ ∈ EQ, CobhamR0 φ Ω₁ (n m : ℕ) : CobhamR0 “↑n + ↑m = ↑(n + m)” Ω₂ (n m : ℕ) : CobhamR0 “↑n * ↑m = ↑(n * m)” Ω₃ (n m : ℕ) : n ≠ m → CobhamR0 “↑n ≠ ↑m” Ω₄ (n : ℕ) : CobhamR0 “∀ x, x < ↑n ↔ ∀ i < n, x = ↑i”</pre>

3.2 (形式的) 証明のルーチン

何らかの算術の体系 T からある論理式を証明できることを証明したい.

$$\text{Lean} \vdash “T \vdash \varphi”$$

しかし, そもそも Lean が形式化された数学であり, その内部でさらに形式化された証明を証明するのはかなり煩雑になる (さらに後には「形式化された形式化された形式化された証明」のようなものさえ扱うことになる).

これは大変なので, 完全性定理を用いて代わりに意味論帰結を証明する.

$$\text{Lean} \vdash “T \models \varphi”$$

意味論的な議論では, Lean のライブラリに用意された代数学の種々の補題やメタプログラミングや自動証明が利用できるため, より簡単に作業を行うことができる.

3.2 (形式的) 証明のルーチン

まず理論の任意のモデル V を固定する.

```
variable {V : Type*} [ORingStruc V] [V ⊨m* Iopen]
```

- **ORingStruc V**: V が言語 \mathcal{L}_{OR} の構造であることを主張する typeclass.
- **$V \models_m^* \text{Iopen}$** : V が理論 Iopen を満たすことを主張する typeclass.


この仮定のもとで証明を行う. 関数を追加したいならば選択関数を用いる.

```
lemma sqrt_exists_unique (a : V) :
  ∃! x, x * x ≤ a ∧ a < (x + 1) * (x + 1) := by ...
def sqrt (a : V) : V := Classical.choose! (sqrt_exists_unique a)
prefix:75 "√" ⇒ sqrt
@[simp] lemma sqrt_mul_self (a : V) : √(a * a) = a := by ...
```

ただし, 帰納法を適用するために, 定義した関数や関係を含む述語がある算術的階層に含まれることが示せなければならない.

3.3 definability tactic

definability tactic は Lean の述語がある算術的階層に含まれることを（可能なら）自動証明する。



typeclass に登録された関数や関係は自動証明の末尾を示すために使用される。

```

instance exp_definable : Σ₀-Function₁ (Exp.exp : V → V) := by ...
instance length_definable : Σ₀-Function₁ (||·|| : V → V) := by ...
instance dvd_definable : Σ₀-Relation (fun a b : V ↦ a | b) := by ...
instance Language.isSemiterm_definable : Δ₁-Relation L.IsSemiterm := by ...

```

3.4 指数関数

指数関数のグラフ $\text{Exp}(x, y) \Leftrightarrow 2^x = y$ は Σ_0 論理式で定義可能 [3], [4].

$$\begin{aligned} \text{Exp}(a_0, a_1) \equiv & [[a_0 = 0] \wedge [a_1 = 1]] \vee [(\exists x_0 < (((a_1 \cdot a_1) \cdot a_1) \cdot a_1 + 1))(\exists x_1 < (((a_1 \cdot a_1) \cdot a_1) \cdot a_1 + 1))(((\exists x_2 < (x_0 + 1))([0 \not\prec ((1 + 1) + 1) + 1]) \vee [(\exists x_3 < (((1 + 1) + 1) + 1) + 1])[x_0 = \\ & (((((1 + 1) + 1) + 1) \cdot x_2) + x_3))]] \wedge [(((1 + 1) + 1) + 1) \neq 0] \vee [x_2 = 0]] \wedge [(\exists x_3 < (x_2 + 1))([0 \not\prec ((1 + 1) + 1) + 1]) \vee [(\exists x_4 < (((1 + 1) + 1) + 1) + 1])[x_2 = (((1 + 1) + 1) + 1) \cdot x_3 + x_4]]] \wedge [(((1 + 1) + 1) + 1) \neq 0] \vee [x_3 = \\ & 0]] \wedge [(((1 + 1) + 1) + 1) \cdot x_3 \neq x_2] \wedge [(((1 + 1) + 1) + 1) \cdot x_3 \not\prec x_2] \vee [x_2 = (((1 + 1) + 1) + 1) \cdot x_3 + 1]] \wedge [x_2 \not\prec (((1 + 1) + 1) + 1) \cdot x_3] \vee [1 = 0]]] \wedge [(\exists x_2 < (x_1 + 1))([0 \not\prec ((1 + 1) + 1) + 1]) \vee [(\exists x_3 < \\ & (((1 + 1) + 1) + 1) + 1])[x_1 = (((1 + 1) + 1) + 1) \cdot x_2 + x_3]] \wedge [(((1 + 1) + 1) + 1) \neq 0] \vee [x_2 = 0]] \wedge [(\exists x_3 < (x_2 + 1))([0 \not\prec ((1 + 1) + 1) + 1]) \vee [(\exists x_4 < (((1 + 1) + 1) + 1) + 1])[x_2 = (((1 + 1) + 1) + 1) \cdot x_3 + x_4]]] \wedge [(((1 + 1) + 1) + 1) \neq \\ & 0] \vee [x_3 = 0]] \wedge [(((1 + 1) + 1) + 1) \cdot x_3 \neq x_2] \wedge [(((1 + 1) + 1) + 1) \cdot x_3 \not\prec x_2] \vee [x_2 = (((1 + 1) + 1) + 1) \cdot x_3 + (1 + 1)] \wedge [x_2 \not\prec (((1 + 1) + 1) + 1) \cdot x_3] \vee [(1 + 1) = 0]]] \wedge [(\forall x_2 < (a_1 + 1))[x_2 = (1 + 1)] \vee [([0 \not\prec \\ & x_2] \vee [(\exists x_3 < (x_2 + 1))([1 < x_3] \wedge [(\exists x_4 < (x_2 + 1))[x_2 = (x_3 \cdot x_4)] \wedge [(\forall x_4 < (x_3 + 1))[x_3 \neq ((1 + 1) \cdot x_4)]]]] \vee [(\forall x_3 < ((1 + 1) \cdot x_2))([(\exists x_4 < (x_3 + 1))([0 \not\prec 1] \vee [(\exists x_5 < 1)[x_3 = ((1 \cdot x_4) + x_5)])] \wedge [1 \neq 0] \vee [x_4 = \\ & 0]) \wedge [(\forall x_5 < (x_4 + 1))[x_4 \neq ((1 + 1) \cdot x_5)]]]] \vee [(\forall x_4 < (x_3 + 1))([0 < (1 + 1)] \wedge [(\forall x_5 < (1 + 1))[x_3 \neq ((1 + 1) \cdot x_4) + x_5]]] \vee [(1 + 1) = 0] \wedge [x_4 \neq 0]] \vee [(\exists x_5 < (x_4 + 1))[x_4 = ((1 + 1) \cdot x_5)]]] \vee [(\exists x_4 < (x_3 + 1))([0 < \\ & x_4] \wedge [(\forall x_5 < (x_4 + 1))[1 \not\prec x_5] \vee [(\forall x_6 < (x_4 + 1))[x_4 \neq (x_5 \cdot x_6)]]] \vee [(\exists x_6 < (x_5 + 1))[x_5 = ((1 + 1) \cdot x_6)]]] \wedge [([1 + 1] < x_4] \wedge [(\exists x_5 < (x_3 + 1))([0 \not\prec x_4] \vee [(\exists x_6 < x_4)[x_3 = ((x_4 \cdot x_5) + x_6)]]] \wedge [x_4 \neq 0] \vee [x_5 = \\ & 0]) \wedge [(\forall x_6 < (x_5 + 1))[x_5 \neq ((1 + 1) \cdot x_6)]]] \wedge [(\forall x_5 < (x_4 + 1))([x_5 \cdot x_5 \neq x_4] \wedge [x_5 \cdot x_5 \not\prec x_4]) \vee [x_4 \not\prec ((x_5 + 1) \cdot x_5 + 1)]] \vee [((x_5 \cdot x_5) \neq x_4] \vee [(\forall x_6 < (x_3 + 1))([0 < x_5] \wedge [(\forall x_7 < x_5)[x_3 \neq ((x_5 \cdot x_6) + x_7)])] \vee [x_5 = \\ & 0] \wedge [x_6 \neq 0]] \vee [(\exists x_7 < (x_6 + 1))[x_6 = ((1 + 1) \cdot x_7)]]] \vee [((\exists x_5 < (x_4 + 1))([x_5 \cdot x_5 = x_4] \wedge [x_5 \cdot x_5 \not\prec x_4]) \vee [x_4 \not\prec ((x_5 + 1) \cdot x_5 + 1)]) \wedge [x_4 < ((x_5 + 1) \cdot x_5 + 1)]] \wedge [x_4 < ((x_5 + 1) \cdot x_5 + 1)] \wedge [(\exists x_6 < (x_3 + 1)) \wedge [(\exists x_7 < x_5)[x_3 = \\ & ((x_5 \cdot x_6) + x_7)]]] \wedge [x_5 \neq 0] \vee [x_6 = 0]] \wedge [(\forall x_7 < (x_6 + 1))[x_6 \neq ((1 + 1) \cdot x_7)]]] \wedge [(\forall x_5 < (x_3 + 1))([0 < x_4] \wedge [(\forall x_6 < x_4)[x_3 \neq ((x_4 \cdot x_5) + x_6)]]] \vee [x_4 = 0] \wedge [x_5 \neq 0]] \vee [(\exists x_6 < (x_5 + 1))[x_5 = ((1 + 1) \cdot x_6)]]] \vee [(\forall x_4 < \\ & (x_3 + 1))([0 < x_2] \wedge [(\forall x_5 < x_2)[x_3 \neq ((x_2 \cdot x_4) + x_5)]]] \vee [x_2 = 0] \wedge [x_4 \neq 0]] \vee [(\exists x_5 < (x_4 + 1))[x_4 = ((1 + 1) \cdot x_5)]]] \vee [((\exists x_3 < (x_0 + 1))([(\exists x_4 < (x_0 + 1))([0 \not\prec x_2] \vee [(\exists x_5 < x_2)[x_0 = ((x_2 \cdot x_4) + x_5)]]] \wedge [x_2 \neq \\ & 0] \vee [x_4 = 0]) \wedge [(\exists x_5 < (x_4 + 1))([0 \not\prec x_2] \vee [(\exists x_6 < x_2)[x_4 = ((x_2 \cdot x_5) + x_6)]]] \wedge [x_2 \neq 0] \vee [x_5 = 0]) \wedge [(((x_2 \cdot x_5) \neq x_4] \wedge [(x_2 \cdot x_5) \not\prec x_4]) \vee [x_4 = ((x_2 \cdot x_5) + x_3)]]] \wedge [x_4 \not\prec (x_2 \cdot x_5)] \vee [x_3 = 0]]] \wedge [(\exists x_4 < (x_0 + 1))([0 \not\prec \\ & (x_2 \cdot x_2)] \vee [(\exists x_5 < (x_2 \cdot x_2))[x_0 = ((x_2 \cdot x_2) \cdot x_4) + x_5])]] \wedge [((x_2 \cdot x_2) \neq 0] \vee [x_4 = 0]) \wedge [(\exists x_5 < (x_4 + 1))([0 \not\prec (x_2 \cdot x_2)] \vee [(\exists x_6 < x_2)[x_4 = ((x_2 \cdot x_5) + x_6)]]] \wedge [x_2 \neq 0] \vee [x_5 = 0]) \wedge [(((x_2 \cdot x_2) \cdot x_5) \neq \\ & x_4] \wedge [((x_2 \cdot x_2) \cdot x_5) \not\prec x_4]) \vee [x_4 = ((x_2 \cdot x_2) \cdot x_5) + ((1 + 1) \cdot x_3)]]] \wedge [x_4 \not\prec ((x_2 \cdot x_2) \cdot x_5)] \vee [((1 + 1) \cdot x_3) = 0]]] \wedge [(\exists x_3 < (x_1 + 1))([(\exists x_4 < (x_1 + 1))([0 \not\prec x_2] \vee [(\exists x_5 < x_2)[x_1 = ((x_2 \cdot x_4) + x_5)]]] \wedge [x_2 \neq 0] \vee [x_4 = \\ & 0]) \wedge [(\exists x_5 < (x_4 + 1))([0 \not\prec x_2] \vee [(\exists x_6 < x_2)[x_4 = ((x_2 \cdot x_5) + x_6)]]] \wedge [x_2 \neq 0] \vee [x_5 = 0]) \wedge [(((x_2 \cdot x_5) \neq x_4] \wedge [(x_2 \cdot x_5) \not\prec x_4]) \vee [x_4 = ((x_2 \cdot x_5) + x_3)]]] \wedge [x_4 \not\prec (x_2 \cdot x_5)] \vee [x_3 = 0]]] \wedge [(\exists x_4 < (x_1 + 1))([0 \not\prec \\ & (x_2 \cdot x_2)] \vee [(\exists x_5 < (x_2 \cdot x_2))[x_0 = ((x_2 \cdot x_2) \cdot x_4) + x_5])]] \wedge [((x_2 \cdot x_2) \neq 0] \vee [x_4 = 0]) \wedge [(\exists x_5 < (x_4 + 1))([0 \not\prec (x_2 \cdot x_2)] \vee [(\exists x_6 < x_2)[x_4 = ((x_2 \cdot x_2) \cdot x_5) + x_6)]]] \wedge [((x_2 \cdot x_2) \neq 0] \vee [x_5 = 0]) \wedge [(((x_2 \cdot x_2) \cdot x_5) \neq \\ & x_4] \wedge [((x_2 \cdot x_2) \cdot x_5) \not\prec x_4]) \vee [x_4 = ((x_2 \cdot x_2) \cdot x_5) + ((1 + 1) \cdot x_3 + 1)]] \wedge [x_4 \not\prec ((x_2 \cdot x_2) \cdot x_5)] \vee [((1 + 1) \cdot x_3 + 1) = 0]]] \wedge [(\exists x_3 < (x_1 + 1))([(\exists x_4 < (x_1 + 1))([0 \not\prec x_2] \vee [(\exists x_5 < x_2)[x_1 = ((x_2 \cdot x_4) + x_5)]]] \wedge [x_2 \neq \\ & 0] \vee [x_4 = 0]) \wedge [(\exists x_5 < (x_4 + 1))([0 \not\prec x_2] \vee [(\exists x_6 < x_2)[x_4 = ((x_2 \cdot x_5) + x_6)]]] \wedge [x_2 \neq 0] \vee [x_5 = 0]) \wedge [(((x_2 \cdot x_5) \neq x_4] \wedge [(x_2 \cdot x_5) \not\prec x_4]) \vee [x_4 = ((x_2 \cdot x_5) + x_3)]]] \wedge [x_4 \not\prec (x_2 \cdot x_5)] \vee [x_3 = 0]]] \wedge [(\exists x_4 < (x_1 + 1))([0 \not\prec \\ & (x_2 \cdot x_2)] \vee [(\exists x_5 < (x_2 \cdot x_2))[x_0 = ((x_2 \cdot x_2) \cdot x_4) + x_5])]] \wedge [((x_2 \cdot x_2) \neq 0] \vee [x_4 = 0]) \wedge [(\exists x_5 < (x_4 + 1))([0 \not\prec (x_2 \cdot x_2)] \vee [(\exists x_6 < x_2)[x_4 = ((x_2 \cdot x_2) \cdot x_5) + x_6)]]] \wedge [((x_2 \cdot x_2) \neq 0] \vee [x_5 = 0]) \wedge [(((x_2 \cdot x_2) \cdot x_5) \neq \\ & x_4] \wedge [((x_2 \cdot x_2) \cdot x_5) \not\prec x_4]) \vee [x_4 = ((x_2 \cdot x_2) \cdot x_5) + ((1 + 1) \cdot (x_3 \cdot x_3))]]] \wedge [x_4 \not\prec ((x_2 \cdot x_2) \cdot x_5)] \vee [((1 + 1) \cdot (x_3 \cdot x_3)) = 0]]] \wedge [(\exists x_2 < ((a_1 \cdot a_1) + 1))[x_2 \neq (1 + 1)] \wedge [([0 < x_2] \wedge [(\forall x_3 < (x_2 + 1))[1 \not\prec \\ & x_3] \vee [(\forall x_4 < (x_2 + 1))[x_2 \neq (x_3 \cdot x_4)]]] \vee [(\exists x_4 < (x_3 + 1))([x_3 = ((1 + 1) \cdot x_4)]]] \wedge [(\exists x_3 < ((1 + 1) \cdot x_2))([(\forall x_4 < (x_3 + 1))([0 < 1] \wedge [(\forall x_5 < 1)[x_3 \neq ((1 \cdot x_4) + x_5)])] \vee [1 = 0] \wedge [x_4 + x_5])]] \vee [1 = 0] \wedge [x_4 + x_5])]] \vee [x_1 = 0] \wedge [x_2 = 0] \wedge [x_3 = \\ & ((1 + 1) \cdot x_5)]]] \wedge [(\exists x_4 < (x_3 + 1))([0 \not\prec (1 + 1)] \vee [(\exists x_5 < (1 + 1))[x_3 = ((1 + 1) \cdot x_4) + x_5])]] \wedge [((1 + 1) \neq 0] \vee [x_4 = 0]) \wedge [(\forall x_5 < (x_4 + 1))[x_4 \neq ((1 + 1) \cdot x_5)]]] \wedge [(\forall x_4 < (x_3 + 1))([0 \not\prec x_4] \vee [(\exists x_5 < (x_4 + 1))[1 < \\ & x_5] \wedge [(\exists x_6 < (x_4 + 1))[x_4 = (x_5 \cdot x_6)]]] \wedge [(\forall x_6 < (x_5 + 1))[x_5 \neq ((1 + 1) \cdot x_6)]]] \vee [((1 + 1) \not\prec x_4] \vee [((\forall x_5 < (x_3 + 1))([0 < x_4] \wedge [(\forall x_6 < x_4)[x_3 \neq ((x_4 \cdot x_5) + x_6)]]] \vee [x_4 = 0] \wedge [x_5 \neq 0]) \vee [(\exists x_6 < (x_5 + 1))[x_5 = \\ & ((1 + 1) \cdot x_6)]]] \vee [(\exists x_5 < (x_4 + 1))([x_5 \cdot x_5 = x_4] \vee [x_5 \cdot x_5 < x_4]) \wedge [x_4 < ((x_5 + 1) \cdot (x_5 + 1))] \wedge [x_5 \cdot x_5 = x_4] \wedge [(\exists x_6 < (x_4 + 1))([0 \not\prec x_5] \vee [(\exists x_7 < x_5)[x_3 = ((x_5 \cdot x_6) + x_7)]]] \wedge [x_5 \neq 0] \vee [x_6 = 0]) \wedge [(\forall x_7 < \\ & (x_6 + 1))[x_6 \neq ((1 + 1) \cdot x_7)]]] \wedge [(\forall x_5 < (x_4 + 1))([x_5 \cdot x_5 \neq x_4] \wedge [x_5 \cdot x_5 \not\prec x_4]) \vee [x_4 \not\prec ((x_5 + 1) \cdot (x_5 + 1))] \vee [((x_5 \cdot x_5) \neq x_4] \vee [(\forall x_6 < (x_3 + 1))([0 < x_5] \wedge [(\forall x_7 < x_5)[x_3 \neq ((x_5 \cdot x_6) + x_7)])] \vee [x_5 = 0] \wedge [x_6 \neq \\ & 0]) \vee [(\exists x_7 < (x_6 + 1))[x_6 = ((1 + 1) \cdot x_7)]]] \vee [(\exists x_5 < (x_3 + 1))([0 \not\prec x_4] \vee [(\exists x_6 < x_4)[x_3 = ((x_4 \cdot x_5) + x_6)]]] \wedge [x_4 \neq 0] \vee [x_5 = 0]) \wedge [(\forall x_6 < (x_5 + 1))[x_5 \neq ((1 + 1) \cdot x_6)]]] \wedge [(\exists x_4 < (x_3 + 1))([0 \not\prec x_2] \vee [(\exists x_5 < \\ & x_2)[x_3 = ((x_2 \cdot x_4) + x_5)]]] \wedge [x_2 \neq 0] \vee [x_4 = 0]] \wedge [(\forall x_5 < (x_4 + 1))[x_4 \neq ((1 + 1) \cdot x_5)]]] \wedge [(\exists x_3 < (x_0 + 1))([0 \not\prec x_2] \vee [(\exists x_4 < x_2)[x_0 = ((x_2 \cdot x_3) + x_4)]]] \wedge [x_2 \neq 0] \vee [x_3 = 0]] \wedge [(\exists x_4 < (x_3 + 1))([0 \not\prec x_2] \vee [(\exists x_5 < \\ & x_2)[x_3 = ((x_2 \cdot x_4) + x_5)]]] \wedge [x_2 \neq 0] \vee [x_4 = 0]] \wedge [(((x_2 \cdot x_4) \neq x_3] \wedge [(x_2 \cdot x_4) \not\prec x_3]) \vee [x_3 = ((x_2 \cdot x_4) + a_0)] \wedge [x_3 \not\prec (x_2 \cdot x_4)] \vee [a_0 = 0]]] \wedge [(\exists x_3 < (x_1 + 1))([0 \not\prec x_2] \vee [(\exists x_4 < x_2)[x_1 = ((x_2 \cdot x_3) + x_4)]]] \wedge [x_2 \neq \\ & 0] \vee [x_3 = 0]] \wedge [(\exists x_4 < (x_3 + 1))([0 \not\prec x_2] \vee [(\exists x_5 < x_2)[x_3 = ((x_2 \cdot x_4) + x_5)]]] \wedge [x_2 \neq 0] \vee [x_4 = 0]] \wedge [(((x_2 \cdot x_4) \neq x_3] \wedge [(x_2 \cdot x_4) \not\prec x_3]) \vee [x_3 = ((x_2 \cdot x_4) + a_1)]]] \wedge [x_3 \not\prec (x_2 \cdot x_4)] \vee [a_1 = 0]]] \wedge [x_3 \neq 0] \vee [x_4 \neq 0]]] \end{aligned}$$

3.5 遺伝的有限集合

「 x は y の要素」 \iff 「 y を二進数展開したとき x 桁目が 1」 と定義する.

$$x \in y \iff \text{Bit}(x, y) \iff \lfloor y/2^x \rfloor \bmod 2 = 1$$

Ackermann coding によって遺伝的有限集合 V_ω が扱える. $\text{I}\Sigma_1$ のもとで基礎的な集合論が展開できる.

```
theorem finset_comprehension1 {Γ} {P : V → Prop} (hP : Γ-[1]-Predicate P) (a : V) :
  ∃ s < exp a, ∀ i < a, i ∈ s ↔ P i -- (制限された)内包公理図式

theorem sUnion_exists_unique (s : V) :
  ∃! u, ∀ x, (x ∈ u ↔ ∃ t ∈ s, x ∈ t) -- 和集合公理

theorem sigma1_replacement {f : V → V} (hf : Σ1-Function1 f) (s : V) :
  ∃! t, ∀ y, (y ∈ t ↔ ∃ x ∈ s, y = f x) -- (制限された)置換公理図式
```

また有限の関数/有限列も扱える.

```
def IsMapping (m : V) : Prop := ∀ x ∈ domain m, ∃! y, ⟨x, y⟩ ∈ m
def Seq (s : V) : Prop := IsMapping s ∧ ∃ l, domain s = under l
```

3.6 原始再帰

$\mathbb{I}\Sigma_1$ では原始再帰法を用いて関数を定義できる.

定理 3.6.1: $f(\vec{v}), g(\vec{v}, x, z)$ を Σ_1 定義可能な関数とする. このとき, 以下を満たす Σ_1 定義可能関数 $\text{PRec}_{f,g}(\vec{v}, x)$ が存在する.

$$\begin{aligned}\text{PRec}_{f,g}(\vec{v}, 0) &= f(\vec{v}) \\ \text{PRec}_{f,g}(\vec{v}, x + 1) &= g(\vec{v}, x, \text{PRec}_{f,g}(\vec{v}, x))\end{aligned}$$

```
structure Blueprint (k : ℕ) where
  zero :  $\Sigma_1$ -Semisentence (k + 1)
  succ :  $\Sigma_1$ -Semisentence (k + 3)
structure Construction {k : ℕ} (φ : Blueprint k) where
  zero : (Fin k → V) → V
  succ : (Fin k → V) → V → V → V
  zero_defined : DefinedFunction zero p.zero
  succ_defined : DefinedFunction (fun v ↦ succ (v .succ.succ) (v 1) (v 0)) p.succ
...
variable {k : ℕ} {φ : Blueprint k} (c : Construction V φ) (v : Fin k → V)
def Construction.result (u : V) : V
theorem Construction.result_zero :
  c.result v 0 = c.zero v
theorem Construction.result_succ (u : V) :
  c.result v (u + 1) = c.succ v u (c.result v u)
```

3.7 再帰的定義

定理 3.7.1: $\Phi_C(\vec{v}, x)$ をクラス C をパラメータとして取る述語だとする. Φ が以下を満たすならば,

1. 述語 $P(c, \vec{v}, x) := \Phi_{\{z \mid z \in c\}}(\vec{v}, x)$ は Δ_1 定義可能.
2. 単調: $C \subseteq C'$ ならば, $\Phi_C(\vec{v}, x) \implies \Phi_{C'}(\vec{v}, x)$
3. 有限: $\Phi_C(\vec{v}, x)$ ならば, ある m が存在して $\Phi_{\{z \in C \mid z < m\}}(\vec{v}, x)$

次を満たす Σ_1 定義可能な述語 $\text{Fix}_\Phi(\vec{v}, x)$ が存在する.

$$\text{Fix}_\Phi(\vec{v}, x) \iff \Phi_{\{x \mid \text{Fix}_\Phi(\vec{v}, x)\}}(\vec{v}, x)$$

さらに次を満たすならば, $\text{Fix}_\Phi(\vec{v}, x)$ は Δ_1 定義可能でその構造帰納法が証明できる.

4. 強有限: $\Phi_C(\vec{v}, x) \implies \Phi_{\{y \in C \mid y < x\}}(\vec{v}, x)$

Fix_Φ を用いることで帰納的に定義されたクラス（リスト，形式化された項，形式化された論理式，形式化された証明，…）を定義できる．

```

structure Blueprint (k : ℕ) where
  core :  $\Delta_1$ -Semisentence (k + 2)

structure Construction ( $\varphi$  : Blueprint k) where
   $\Phi$  : (Fin k  $\rightarrow$  V)  $\rightarrow$  Set V  $\rightarrow$  V  $\rightarrow$  Prop
  defined : Arith.Defined (fun v  $\mapsto$   $\Phi$  (v ..succ.succ) {x | x  $\in$  v 1} (v 0))  $\varphi$ .core
  monotone {C C' : Set V} (h : C  $\subseteq$  C') {v x} :  $\Phi$  v C x  $\rightarrow$   $\Phi$  v C' x

class Construction.Finite (c : Construction V  $\varphi$ ) where
  finite {C : Set V} {v x} : c. $\Phi$  v C x  $\rightarrow$   $\exists$  m, c. $\Phi$  v {y  $\in$  C | y < m} x

class Construction.StrongFinite (c : Construction V  $\varphi$ ) where
  strong_finite {C : Set V} {v x} : c. $\Phi$  v C x  $\rightarrow$  c. $\Phi$  v {y  $\in$  C | y < x} x
  ...

variable {k : ℕ} { $\varphi$  : Blueprint k} (c : Construction V  $\varphi$ ) [Finite c] (v : Fin k  $\rightarrow$  V)

def Construction.Fixpoint (x : V) : Prop
theorem Construction.case :
  c.Fixpoint v x  $\leftrightarrow$  c. $\Phi$  v {z | c.Fixpoint v z} x

theorem induction [c.StrongFinite] {P : V  $\rightarrow$  Prop} (hP :  $\Gamma$ -[1]-Predicate P)
  (H :  $\forall$  C : Set V, ( $\forall$  x  $\in$  C, c.Fixpoint v x  $\wedge$  P x)  $\rightarrow$   $\forall$  x, c. $\Phi$  v C x  $\rightarrow$  P x) :
   $\forall$  x, c.Fixpoint v x  $\rightarrow$  P x

```

4. メタ数学の算術化

4.1 項のコーディング

Δ_1 述語 T_C を以下のように定める.

$$u \in T_C \iff (\exists z) [u = \widehat{\#z}] \vee (\exists x) [u = \widehat{\&x}] \vee \\ (\exists k, f, v) \left[\text{Func}(k, f) \wedge \text{len}(v) = k \wedge (\forall i < k) [\text{nth}(v, i) \in C] \wedge u = \widehat{f^k(v)} \right]$$

$$\text{bounded variable: } \widehat{\#z} := \langle 0, z \rangle + 1$$

$$\text{free variable: } \widehat{\&z} := \langle 1, z \rangle + 1$$

$$\text{function: } \widehat{f^k(v)} := \langle 2, f, k, v \rangle + 1$$

\mathcal{L} を形式化された言語だとする. fixpoint construction により T_C の不動点 $\mathcal{L}.\text{IsUTerm} : V \rightarrow \text{Prop}$ が得られる. また, その構造帰納法から束縛変数の最大値+1 を返す Σ_1 関数 $\mathcal{L}.\text{termBV} : V \rightarrow V$ が定義でき, $t \in V$ がコード化された (n 個の束縛変数を持つ) 擬項であることを表す Δ_1 述語 $\mathcal{L}.\text{IsSemiterm } n \ t : \text{Prop}$ が定義される.

```
def Language.IsSemiterm (n t : V) : Prop := L.IsUTerm t ∧ L.termBV t ≤ n
```

4.2 論理式のコーディング

Δ_1 述語 F_C を以下のように定める.

$$\begin{aligned}
 u \in F_C &\iff (\exists k, R, v) \left[\text{Rel}(k, R) \wedge u \text{ is a list of UTerm of length } k \wedge u = \widehat{R^k(v)} \right] \vee \\
 &(\exists k, R, v) \left[\text{Rel}(k, R) \wedge u \text{ is a list of UTerm of length } k \wedge u = \neg \widehat{R^k(v)} \right] \vee \\
 &u = \hat{\top} \vee u = \hat{\perp} \vee \\
 &(\exists p, q \in C) \left[u = \widehat{p \wedge q} \right] \vee (\exists p, q \in C) \left[u = \widehat{p \vee q} \right] \vee \\
 &(\exists p \in C) \left[u = \widehat{\forall p} \right] \vee (\exists p \in C) \left[u = \widehat{\exists p} \right]
 \end{aligned}$$

$$\begin{aligned}
 \widehat{R^k(v)} &:= \langle 0, R, k, v \rangle + 1 & \hat{\top} &:= \langle 2, 0 \rangle + 1 & \widehat{p \wedge q} &:= \langle 4, p, q \rangle + 1 & \widehat{\forall p} &:= \langle 6, p \rangle + 1 \\
 \neg \widehat{R^k(v)} &:= \langle 1, R, k, v \rangle + 1 & \hat{\perp} &:= \langle 3, 0 \rangle + 1 & \widehat{p \vee q} &:= \langle 5, p, q \rangle + 1 & \widehat{\exists p} &:= \langle 7, p \rangle + 1
 \end{aligned}$$

擬項のときと同様にして形式化された擬論理式を指す Δ_1 述語 $\text{IsSemiformula } n \ p : \text{Prop}$ が定義される.

4.3 Tait 計算のコーディング

T を Δ_1 定義可能な理論とする. Δ_1 述語 D_C^T を以下のように定める.

$$\begin{aligned}
 d \in D_C^T \iff & (\forall p \in \text{sq}(d))[\text{Semiformula}(0, p)] \wedge \\
 & [(\exists s, p)[d = \text{AXL}(s, p) \wedge p \in s \wedge \hat{\neg} p \in s] \\
 & (\exists s)[d = \top\text{-INTRO}(s) \wedge \hat{\top} \in s] \vee \\
 & (\exists s, p, q)(\exists d_p, d_q \in C)[d = \wedge\text{-INTRO}(s, p, q, d_p, d_q) \wedge \widehat{p \wedge q} \in s \wedge \text{sq}(d_p) = s \cup \{p\} \wedge \text{sq}(d_q) = s \cup \{q\}] \vee \\
 & (\exists s, p, q)(\exists d \in C)[d = \vee\text{-INTRO}(s, p, q, d) \wedge \widehat{p \vee q} \in s \wedge \text{sq}(d) = s \cup \{p, q\}] \vee \\
 & (\exists s, p)(\exists d \in C)[d = \forall\text{-INTRO}(s, p, d) \wedge \widehat{\forall p} \in s \wedge \text{sq}(d) = s^+ \cup \{p^+(\widehat{\&0})\}] \vee \\
 & (\exists s, p, t)(\exists d \in C)[d = \exists\text{-INTRO}(s, p, t, d) \wedge \widehat{\exists p} \in s \wedge \text{sq}(d) = s \cup \{p(t)\}] \vee \\
 & (\exists s)(\exists d' \in C)[d = \text{WK}(s, d') \wedge s \supseteq \text{sq}(d')] \vee \\
 & (\exists s)(\exists d' \in C)[d = \text{SHIFT}(s, d') \wedge s = \text{sq}(d')^+] \vee \\
 & (\exists s, p)(\exists d_1, d_2 \in C)[d = \text{CUT}(s, p, d_1, d_2) \wedge \text{sq}(d_1) = s \cup \{p\} \wedge \text{sq}(d_2) = s \cup \{\hat{\neg} p\}] \vee \\
 & (\exists s, p)[d = \text{ROOT}(s, p) \wedge p \in s \wedge p \in T]
 \end{aligned}$$

$$\begin{aligned}
& \top\text{-INTRO}(s) := \langle s, 1, 0 \rangle + 1 & \text{WK}(s, d) &:= \langle s, 6, d \rangle + 1 \\
& \text{sqt}(d) := \pi_1(d - 1) & \wedge\text{-INTRO}(s, p, q, d_p, d_q) &:= \langle s, 2, p, q, d_p, d_q \rangle + 1 & \text{SHIFT}(s, d) &:= \langle s, 7, d \rangle + 1 \\
& \text{AXL}(s, p) := \langle s, 0, p \rangle + 1 & \vee\text{-INTRO}(s, p, q, d) &:= \langle s, 3, p, q, d \rangle + 1 & \text{CUT}(s, p, d_1, d_2) &:= \langle s, 8, p, d_1 d_2 \rangle + 1 \\
& & \forall\text{-INTRO}(s, p, d) &:= \langle s, 4, p, d \rangle + 1 & \text{ROOT}(s, p) &:= \langle s, 9, p \rangle + 1 \\
& & \exists\text{-INTRO}(s, p, t, d) &:= \langle s, 5, p, t, d \rangle + 1
\end{aligned}$$

D_C^T の不動点を取って $\top.\text{Derivation}$ とする。以下のように定める。

```

def Language.Theory.Derivable (T) (s : V) : Prop := ∃ d, T.DerivationOf d s
def Language.Theory.Provable (T) (p : V) : Prop := T.Derivable {p}

```

定義より $\top.\text{Derivable}$ や $\top.\text{Provable}$ は Σ_1 .

4.4 Gödel 数化

(メタ論理の) 項/論理式から (V の内部の) 形式化された項/論理式への翻訳 $t \mapsto [t], \varphi \mapsto [\varphi]$ をコーディング通りに定義する.

```
#eval Encodable.encode (“⊢” : Sentence  $\mathcal{L}_{\text{or}}$ )
/- 7 -/

#eval Encodable.encode (“1 + 1 = 2” : Sentence  $\mathcal{L}_{\text{or}}$ )
/- 2811283025421999017712752184705287682765933652183125347889924839244702557909257480149303662
0226056829379396867558990018685021300411793862047857546098162625163543635122202320614830504
5032994237166744576048470977246790815309324291777191455007248230247852452218972967918789604
6549123727569857552482425621934531125928101852294893549785680809310809523176634584145844302
5403080631492029306853223861326740190114231247862748245410665797364948572233034830050682843
0586068980626109306946946510933159275481055524034583309159054488191374946941436375286182247
2934435856999319941455469415760760469554147547207581743524630022634897251631217278137049000
0720142817763639758008432269310312579231167553225000584223488849283104065326226084742908739
1926282639848319450229266995417413133859040482530307149746254214536136234612283878168584544
0140202739452286492522894832319502650575282054270818964898878686030069994727438159066748780
6793436703922580916715270838972818987012192890409519287109955207836968510432355251110156455
5848127363401105585258109417133714744133195894888409659553672360357137318621009961116974146
9010059357928982517174650081907896565063353817595634565097605705845515007215881579535384687
7165987904543322149860791463967564337677539451591793909705364343215133849384526734310253600
618529011635630072568447447378714105148191145285 -/
```

5. Hilbert-Bernays-Löb の可証性条件

5.1 可証性条件

補題 5.1.1: T を $\text{I}\Sigma_1$ より強い理論, U を R_0 より強い Δ_1 定義可能な理論だとする. このとき,

D1 $U \vdash \sigma \implies T \vdash \text{Provable}_U(\ulcorner \sigma \urcorner)$

D2 $T \vdash \text{Provable}_U(\ulcorner \sigma \rightarrow \tau \urcorner) \rightarrow \text{Provable}_U(\ulcorner \sigma \urcorner) \rightarrow \text{Provable}_U(\ulcorner \tau \urcorner)$

D3 $T \vdash \text{Provable}_U(\ulcorner \sigma \urcorner) \rightarrow \text{Provable}_U(\ulcorner \text{Provable}_U(\ulcorner \sigma \urcorner) \urcorner)$

$\mathbb{N} \models T$ ならば¹ 更に次が成り立つ.

D1' $U \vdash \sigma \iff T \vdash \text{Provable}_U(\ulcorner \sigma \urcorner)$

¹実際には Σ_1 健全性で十分.

5.1 可証性条件

D1 及び D2, D1' は形式化された証明の性質を地道に証明すれば示せる.

theorem **provable_a_D1**

$[I\Sigma_1 \preceq T] [U.\text{Delta1Definable}] \{\sigma : \text{SyntacticFormula } L\} :$
 $U \vdash! \sigma \rightarrow T \vdash! U.\text{bew}_a \sigma$

theorem **provable_a_D2**

$[I\Sigma_1 \preceq T] [U.\text{Delta1Definable}] \{\sigma \pi : \text{SyntacticFormula } L\} :$
 $T \vdash! U.\text{bew}_a (\sigma \rightarrow \pi) \rightarrow U.\text{bew}_a \sigma \rightarrow U.\text{bew}_a \pi$

theorem **provable_a_complete**

$[I\Sigma_1 \preceq T] [N \models_m^* T] [R_0 \preceq U] \{\sigma : \text{LO.FirstOrder.Sentence } \mathcal{L}_{or}\} :$
 $U \vdash! \sigma \leftrightarrow T \vdash! U.\text{bew}_a \sigma$

12

¹ $T \vdash \varphi$ は φ の理論 T -証明の型, $T \vdash! \varphi$ はそのような証明が存在することを表す命題.

² $T.\text{Provable } p$ は $p \in V$ に関するモデル上の形式化された証明可能性を表す. 一方 $T.\text{bew}_a \sigma$ は $T.\text{Provable } \ulcorner \sigma \urcorner$ を定義する論理式を指す.

5.2 形式化された Σ_1 -完全性

D3 は直接示すのは難しいが、次の補題より従う.

補題 5.2.1 (形式化された Σ_1 -完全性): T を $\text{I}\Sigma_1$ より強い理論, U を R_0 より強い Δ_1 定義可能な理論だとする. 文 σ が Σ_1 論理式ならば, 次が証明可能.

$$T \vdash \sigma \rightarrow \text{Provable}_U(\ulcorner \sigma \urcorner)$$

証明 T のモデル V の内部で作業する.

次を示せばよい: すべての Σ_1 論理式 $\varphi(x_1, \dots, x_k)$, $a_1, \dots, a_k \in V$ について, $V \models \varphi[a_1, \dots, a_k] \implies \text{Provable}_U(\ulcorner \varphi(\overline{a_1}, \dots, \overline{a_k}) \urcorner)$ これは φ に関する帰納法で示せる. \square

theorem `provablea-sigma1-complete`

$\{\sigma : \text{Sentence } \mathcal{L}_{\text{or}}\}$ ($\text{h}\sigma : \text{Hierarchy } \Sigma \text{ 1 } \sigma$) :
 $T \vdash! \sigma \rightarrow U.\text{bew}_a \sigma$

6. 不完全性定理

6.1 第一不完全性定理

G1 は 1, 2, 3, 4 までの結果で証明できる.

定理 6.1.1 (G1): T が Δ_1 定義可能で R_0 より強く Σ_1 -健全ならば, T から証明も反証もできない論理式が存在する.

証明 $D := \{[\varphi] \mid \varphi : 1 \text{ 変数の論理式}, T \vdash \neg\varphi([\varphi])\}$ と定義する. $T \vdash \pi \Leftrightarrow \mathbb{N} \models \text{Provable}_T([\pi])$, また Provable_T は Σ_1 定義可能なので, D は r.e. 集合.

表現定理より次を満たす θ が存在する: すべての $n \in \mathbb{N}$ について $n \in D \Leftrightarrow T \vdash \theta(\bar{n})$.

$n = [\theta]$ と置くと,

$$T \vdash \theta([\theta]) \Leftrightarrow [\theta] \in D \Leftrightarrow T \vdash \neg\theta([\theta])$$

よって T が完全だと仮定すると矛盾する. \square

6.1 第一不完全性定理

```

theorem goedel_first_incompleteness : ¬System.Complete T := by
  let D : ℕ → Prop := fun n : ℕ ↦ ∃ p : SyntacticSemiformula L0, 1, n = ⌈p⌉ ∧ T ⊢! ¬p/⌈p⌉
  -- D の定義
  have D_re : RePred D := by ...
  -- D は r.e.
  let θ : SyntacticSemiformula L0, 1 := codeOfRePred (D)
  have : ∀ n : ℕ, D n ↔ T ⊢! θ/⌈↑n⌉ := fun n ↦ by
    simpa [Semiformula.coe_substs_eq_substs_coe1] using re_complete (T := T) (D_re) (x := n)
  -- D を表現する論理式 θ を取る
  let ρ : SyntacticFormula L0 := θ/⌈↑θ⌉
  -- ρ := θ(↑θ)
  have : T ⊢! ¬ρ ↔ T ⊢! ρ := by
    simpa [D, goedelNumber'_def, quote_eq_encode] using this ⌈↑θ⌉
  -- T ⊢ ¬ρ ↔ T ⊢ ρ
  have con : System.Consistent T := consistent_of_sigma1Sound T
  -- T は無矛盾
  refine LO.System.incomplete_iff_exists_undecidable.mpr ⟨↑ρ, ?_, ?_⟩
  -- ρ が T から独立であることを示す
  · intro h
    have : T ⊢! ¬↑ρ := by simpa [provableθ_iff] using this.mpr h
    exact LO.System.not_consistent_iff_inconsistent.mpr (inconsistent_of_provable_of_unprovable h this) inferInstance
    -- T ⊢ ¬ρ ならば矛盾.
  · intro h
    have : T ⊢! ↑ρ := this.mp (by simpa [provableθ_iff] using h)
    exact LO.System.not_consistent_iff_inconsistent.mpr (inconsistent_of_provable_of_unprovable this h) inferInstance
    -- T ⊢ ρ ならば矛盾.

```

6.2 第二不完全性定理

補題 6.2.1 (不動点補題): 1 変数の論理式 θ について, 次を満たす 文 fixpoint_θ が存在する.

$$|\Sigma_1 \vdash \text{fixpoint}_\theta \leftrightarrow \theta(\lceil \text{fixpoint}_\theta \rceil)$$

証明 形式化された論理式に数項を代入する関数のグラフ $\text{substNumeal}(y, p, x)$ は Σ_1 定義可能.

$$|\Sigma_1 \vdash \text{substNumeal}(y, p, x) \leftrightarrow y = p(\bar{x})$$

fixpoint_θ を次のように定義する.

$$\text{diag}_\theta(x) := (\forall y)[\text{substNumeal}(y, x, x) \rightarrow \theta(y)]$$

$$\text{fixpoint}_\theta := \text{diag}_\theta(\lceil \text{diag}_\theta \rceil)$$

このとき, $|\Sigma_1$ に於いて,

$$\begin{aligned} \text{fixpoint}_\theta &\stackrel{\text{def}}{=} (\forall y)[\text{substNumeal}(y, \lceil \text{diag}_\theta \rceil, \lceil \text{diag}_\theta \rceil) \rightarrow \theta(y)] \\ &\leftrightarrow \theta(\lceil \text{diag}_\theta(\lceil \text{diag}_\theta \rceil) \rceil) \\ &\stackrel{\text{def}}{=} \theta(\lceil \text{fixpoint}_\theta \rceil) \end{aligned}$$

□

6.2 第二不完全性定理

Gödel 文 「この文は証明できない」 を定義する. $G_T := \text{fixpoint}_{\neg \text{Provable}_T}$

以降 T を $\text{I}\Sigma_1$ より強い理論とする.

補題 6.2.2:

1. T が無矛盾ならば $T \not\vdash G_T$,
2. $\mathbb{N} \models T$ ならば $T \not\vdash \neg G_T$.

証明

1. $T \vdash G_T$ と仮定する. 不動点補題より $T \vdash \neg \text{Provable}_T(\ulcorner G_T \urcorner)$ を得る. 一方 D1 より $T \vdash \text{Provable}_T(\ulcorner G_T \urcorner)$ なので T は矛盾する. これは仮定に反する.
2. $T \vdash \neg G_T$ と仮定する. 不動点補題より $T \vdash \text{Provable}_T(\ulcorner G_T \urcorner)$ を得る. D1' より $T \vdash G_T$ なので T は矛盾する. これは仮定に反する.

6.2 第二不完全性定理

T の無矛盾性を表す文を定義する. $\text{Con}_T := \neg \text{Provable}_T(\ulcorner \perp \urcorner)$

補題 6.2.3: $T \vdash \text{Con}_T \leftrightarrow G_T$

証明

(\rightarrow) $\neg G_T \rightarrow \text{Provable}_T(\ulcorner \perp \urcorner)$ を示せばよい. $\neg G_T$ を仮定する. 不動点補題より $\text{Provable}_T(\ulcorner G_T \urcorner)$. また, 不動点補題と D1 より $\text{Provable}_T(\ulcorner G_T \rightarrow \neg \text{Provable}_T(\ulcorner G_T \urcorner) \urcorner)$. よって D2 より $\text{Provable}_T(\ulcorner \neg \text{Provable}_T(\ulcorner G_T \urcorner) \urcorner)$.

一方 D3 より $\text{Provable}_T(\ulcorner \text{Provable}_T(\ulcorner G_T \urcorner) \urcorner)$. 再び D2 を用いて $\text{Provable}_T(\ulcorner \perp \urcorner)$ を得る.

(\leftarrow) $\text{Provable}_T(\ulcorner \perp \urcorner) \rightarrow \neg G_T$ を示せばよい. $\text{Provable}_T(\ulcorner \perp \urcorner)$ を仮定する. D1 より $\text{Provable}_T(\ulcorner \perp \rightarrow G_T \urcorner)$, よって D2 より $\text{Provable}_T(\ulcorner G_T \urcorner)$. 不動点補題より $\neg G_T$ を得る. \square

定理 6.2.1 (G2): T が無矛盾ならば $T \not\vdash \text{Con}_T$. $\mathbb{N} \models T$ ならば $T \not\vdash \neg \text{Con}_T$.

6.2 第二不完全性定理

よって以下が証明できる.

```
theorem goedel_second_incompleteness
  [IΣ1 ≤ T] [T.Delta1Definable] [LO.System.Consistent T] :
  T ⊄ ↑Con
```

```
theorem inconsistent_undecidable
  [IΣ1 ≤ T] [T.Delta1Definable] [N ⊨m* T] :
  System.Undecidable T ↑Con
```

ただし, 上の証明が本質的に依存しているのは D1, D2, D3 と述語の健全性のみ.

6.3 証明可能性論理

以下の内容は私（齋藤）ではなく，主に神戸大の野口氏の形式化した結果である．

型クラスを用いて証明可能性述語や Hilbert-Bernays の可証性条件を抽象的に定義できる．

```

structure ProvabilityPredicate (T : Theory L) (U : Theory L) where
  prov : Semisentence L 1
  spec {σ : Sentence L} : U ⊢! σ → T ⊢! prov/[「σ」] -- 少なくとも D1 を満たす

class Diagonalization (T : Theory L) where
  fixpoint : Semisentence L 1 → Sentence L
  diag (θ) : T ⊢!. fixpoint θ ↔ θ/[「fixpoint θ」]

...

class ProvabilityPredicate.HBL2 (ℬ : ProvabilityPredicate T U) where
  D2 {σ τ : Sentence L} : T ⊢! ℬ (σ → τ) → (ℬ σ) → (ℬ τ) -- D2

class ProvabilityPredicate.HBL3 (ℬ : ProvabilityPredicate T U) where
  D3 {σ : Sentence L} : T ⊢! (ℬ σ) → ℬ (ℬ σ) -- D3

class ProvabilityPredicate.HBL (ℬ : ProvabilityPredicate T₀ T)
  extends ℬ.HBL2, ℬ.HBL3 -- D1 + D2 + D3

```

6.3 証明可能性論理

これらの仮定の上で一般的に $G1$ や $G2$ が証明できる.

第一不完全性定理：

```
theorem goedel_independent
  [T ≤ U] [Diagonalization T] [LO.System.Consistent U]
  (ℬ : ProvabilityPredicate T U) [ℬ.GoedelSound] :
  System.Undecidable U (goedel ℬ)
```

第二不完全性定理：

```
theorem unprovable_consistency
  [T ≤ U] [Diagonalization T] [System.Consistent U]
  (ℬ : ProvabilityPredicate T U) [ℬ.HBL] :
  U ≠ con ℬ
```

6.4 今後

- 算術的完全性定理.
- Paris-Harrington の定理等の独立命題に関する結果.
- 集合論.
- 二階算術.
- 直観主義一階述語論理, 特に Heyting arithmetic.
- Büchi arithmetic や S2S の決定性.
- 自動証明.
-

Bibliography

- [1] L. C. Paulson, “A mechanised proof of Gödel’s incompleteness theorems using Nominal Isabelle,” *Journal of Automated Reasoning*, vol. 55, pp. 1–37, 2015.
- [2] W. Pohlers, *Proof theory: The first step into impredicativity*. Springer Science & Business Media, 2008.
- [3] P. Hájek and P. Pudlák, *Metamathematics of first-order arithmetic*, vol. 3. Cambridge University Press, 2017.
- [4] S. Cook and P. Nguyen, *Logical foundations of proof complexity*, vol. 11. Cambridge University Press Cambridge, 2010.

Sponsor

FormalizedFormalLogic  is supported by Proxima Technology  **Proxima**